AMS-512 Capital Markets and Portfolio Theory

Studying Market Draw Downs

Robert J. Frey, Research Professor
Stony Brook University, Applied Mathematics and Statistics

frey@ams.sunysb.edu
http://www.ams.sunysb.edu/~frey

Market Draw Downs

Market draw downs do not follow a simple stochastic process. Even though there is little or not detectable serial correlation in returns, the occurrence of large market draw downs is much more likely than expected. This indicates that there is some form of non-stationarity, probably in the form of regime changes. How we reflect these effects in projecting future risks and deciding upon acceptable levels of risk in constructing portfolios is a more complicated process than it first appears.

Set-Up

Local Save

This function constructs a full path to export files into the same directory as the notebook.

```
In[]:= sHere[name_] := FileNameJoin[{NotebookDirectory[], name}];

In[]:= sHere["file.m"]

Out[]= /Users/robertjfrey/Documents/Work/Stony Brook University/AMS/QF/CourseWork/AMS-512/Modules/DrawDowns/file.m
```

Data

The file “mxSP500TRIndex.m” contains almost two centuries of monthly returns for the S&P 500 Total Return Index, *i.e.*, with dividends reinvested.

In[∘]:=
```
mxSP500TRIndex = Import[FileNameJoin[{NotebookDirectory[], "mxSP500TRIndex.m"}]];
```

In[∘]:=
```
TableForm@mxSP500TRIndex⟦1 ;; 3⟧
```

Out[∘]//TableForm=
```
1835
1        -6.65249
31
1835
2        -6.66393
28
1835
3        -6.59748
31
```

In[∘]:=
```
TableForm@mxSP500TRIndex⟦-3 ;;⟧
```

Out[∘]//TableForm=
```
2018
2        8.57699
28
2018
3        8.55124
29
2018
4        8.53765
6
```

To aid in plotting we extract the first and last months’ dates and format them into a readable format.

In[∘]:=
```
sFromDate = DateString[First[mxSP500TRIndex⟦All, 1⟧], {"MonthNameShort", " ", "Day", ", ", "Year"}]
sUntilDate = DateString[Last[mxSP500TRIndex⟦All, 1⟧], {"MonthNameShort", " ", "Day", ", ", "Year"}]
```

Out[∘]= Jan 31, 1835

Out[∘]= Apr 06, 2018

Below is a plot of the log index with draw downs colored in using the Filing option in Plot[ ].

In[ ]:=
```
Show[
 DateListPlot[
  {
   mxSP500TRIndex,
   Transpose[{First /@ #, Rest[FoldList[Max, -∞, Last /@ #]]}] &[mxSP500TRIndex]
  },
  Joined → True,
  PlotStyle → {Black, Black},
  Filling → {1 → {{2}, Opacity[0.5, Red]}},
  Frame → True,
  FrameLabel → {
    {"Log Index", ""},
    {
     "Date",
     Column[
      {
       Style["Market Draw Downs " <> sFromDate <> " through " <> sUntilDate, FontSize → 14],
       Style["S&P 500 Total Return Index", FontSize → 14],
       Style["Source: Global Financial Data", FontSize → 10]
      },
      Center
     ]
    }
   },
  ImageSize → 500,
  ImageMargins → 10
 ]
]
```

Out[•]=



Drawdown — All Data

Extracting Draw Downs

We can construct the draw downs, expressed as positive numbers, by taking the running maximum of the log index and subtracting the log index from it.

In[•]:=

```
mxDrawdowns[All] = Transpose[{mxSP500TRIndex〚All, 1〛, Rest[FoldList[Max, -∞, #]] - # &[mxSP500TRIndex〚All, 2〛]}];
```

In[•]:=
```
Show[
 DateListPlot[
  mxDrawdowns[All],
  Joined → True,
  PlotRange → All,
  PlotStyle → Black,
  Filling → Bottom,
  FillingStyle → Opacity[0.5, Red],
  GridLines → {Table[{i, 1, 1}, {i, 1840, 2000, 10}], Table[i, {i, 0, 2, 0.1}]},
  FrameLabel → {
    {"Log Draw Down", ""},
    {
     "Date",
     Column[
      {
       Style["Market Draw Downs " <> sFromDate <> " through " <> sUntilDate, FontSize → 14],
       Style["S&P 500 Total Return Index", FontSize → 14],
       Style["Source: Global Financial Data", FontSize → 10]
      },
      Center
     ]
    }
   }
 ],
 ImageSize → 500,
 ImageMargins → 10
]
```

Out[•]=



Next, we use Split[ ] to break the draw downs into interspersed no draw down and draw down intervals.

In[•]:=
```
mxDrawdowns[All]⟦1 ;; 20⟧
```

Out[•]=
```
{{{1835, 1, 31}, 0.}, {{1835, 2, 28}, 0.0114342}, {{1835, 3, 31}, 0.}, {{1835, 4, 30}, 0.},
 {{1835, 5, 31}, 0.}, {{1835, 6, 30}, 0.00374822}, {{1835, 7, 31}, 0.0129369}, {{1835, 8, 31}, 0.},
 {{1835, 9, 30}, 0.0428973}, {{1835, 10, 31}, 0.059996}, {{1835, 11, 30}, 0.122066}, {{1835, 12, 31}, 0.139146},
 {{1836, 1, 31}, 0.126248}, {{1836, 2, 29}, 0.0403939}, {{1836, 3, 31}, 0.0391106}, {{1836, 4, 30}, 0.0700374},
 {{1836, 5, 31}, 0.0690301}, {{1836, 6, 30}, 0.0680029}, {{1836, 7, 31}, 0.102414}, {{1836, 8, 31}, 0.130283}}
```

In[•]:=
```
vxIntervals[All] = Split[mxDrawdowns[All], ((#1⟦2⟧ > 0 && #2⟦2⟧ > 0) || (#1⟦2⟧ == 0 && #2⟦2⟧ == 0)) &];
```

```
In[•]:= vxIntervals[All]⟦1 ;; 7⟧

Out[•]= {{{{1835, 1, 31}, 0.}}, {{{1835, 2, 28}, 0.0114342}}, {{{1835, 3, 31}, 0.}, {{1835, 4, 30}, 0.}, {{1835, 5, 31}, 0.}},
  {{{1835, 6, 30}, 0.00374822}, {{1835, 7, 31}, 0.0129369}}, {{{1835, 8, 31}, 0.}},
  {{{1835, 9, 30}, 0.0428973}, {{1835, 10, 31}, 0.059996}, {{1835, 11, 30}, 0.122066}, {{1835, 12, 31}, 0.139146},
   {{1836, 1, 31}, 0.126248}, {{1836, 2, 29}, 0.0403939}, {{1836, 3, 31}, 0.0391106}, {{1836, 4, 30}, 0.0700374},
   {{1836, 5, 31}, 0.0690301}, {{1836, 6, 30}, 0.0680029}, {{1836, 7, 31}, 0.102414}, {{1836, 8, 31}, 0.130283},
   {{1836, 9, 30}, 0.157049}, {{1836, 10, 31}, 0.211886}, {{1836, 11, 30}, 0.2395}, {{1836, 12, 31}, 0.204301},
   {{1837, 1, 31}, 0.280232}, {{1837, 2, 28}, 0.308122}, {{1837, 3, 31}, 0.350551}, {{1837, 4, 30}, 0.387401},
   {{1837, 5, 31}, 0.372081}, {{1837, 6, 30}, 0.307865}, {{1837, 7, 31}, 0.294654}, {{1837, 8, 31}, 0.234043},
   {{1837, 9, 30}, 0.212047}, {{1837, 10, 31}, 0.269299}, {{1837, 11, 30}, 0.253948}, {{1837, 12, 31}, 0.266833},
   {{1838, 1, 31}, 0.246464}, {{1838, 2, 28}, 0.275124}, {{1838, 3, 31}, 0.307643}, {{1838, 4, 30}, 0.319302},
   {{1838, 5, 31}, 0.260141}, {{1838, 6, 30}, 0.196289}, {{1838, 7, 31}, 0.201184}, {{1838, 8, 31}, 0.149617},
   {{1838, 9, 30}, 0.13038}, {{1838, 10, 31}, 0.18108}, {{1838, 11, 30}, 0.173554}, {{1838, 12, 31}, 0.19118},
   {{1839, 1, 31}, 0.151081}, {{1839, 2, 28}, 0.114582}, {{1839, 3, 31}, 0.15083}, {{1839, 4, 30}, 0.138254},
   {{1839, 5, 31}, 0.108701}, {{1839, 6, 30}, 0.130313}, {{1839, 7, 31}, 0.165637}, {{1839, 8, 31}, 0.189278},
   {{1839, 9, 30}, 0.208375}, {{1839, 10, 31}, 0.253578}, {{1839, 11, 30}, 0.292099}, {{1839, 12, 31}, 0.26031},
   {{1840, 1, 31}, 0.21536}, {{1840, 2, 29}, 0.224474}, {{1840, 3, 31}, 0.251134}, {{1840, 4, 30}, 0.22307},
   {{1840, 5, 31}, 0.217848}, {{1840, 6, 30}, 0.206978}, {{1840, 7, 31}, 0.199064}, {{1840, 8, 31}, 0.208191},
   {{1840, 9, 30}, 0.177692}, {{1840, 10, 31}, 0.120865}, {{1840, 11, 30}, 0.126522}, {{1840, 12, 31}, 0.145941},
   {{1841, 1, 31}, 0.166168}, {{1841, 2, 28}, 0.169672}, {{1841, 3, 31}, 0.241766}, {{1841, 4, 30}, 0.209501},
   {{1841, 5, 31}, 0.160529}, {{1841, 6, 30}, 0.155546}, {{1841, 7, 31}, 0.139161}, {{1841, 8, 31}, 0.148496},
   {{1841, 9, 30}, 0.178557}, {{1841, 10, 31}, 0.197751}, {{1841, 11, 30}, 0.183837}, {{1841, 12, 31}, 0.228574},
   {{1842, 1, 31}, 0.370359}, {{1842, 2, 28}, 0.38766}, {{1842, 3, 31}, 0.425022}, {{1842, 4, 30}, 0.416297},
   {{1842, 5, 31}, 0.343322}, {{1842, 6, 30}, 0.302508}, {{1842, 7, 31}, 0.337142}, {{1842, 8, 31}, 0.36204},
   {{1842, 9, 30}, 0.334759}, {{1842, 10, 31}, 0.344473}, {{1842, 11, 30}, 0.373786}, {{1842, 12, 31}, 0.368839},
   {{1843, 1, 31}, 0.387065}, {{1843, 2, 28}, 0.33878}, {{1843, 3, 31}, 0.333538}, {{1843, 4, 30}, 0.320614},
   {{1843, 5, 31}, 0.197846}, {{1843, 6, 30}, 0.140414}, {{1843, 7, 31}, 0.144741}, {{1843, 8, 31}, 0.107843},
   {{1843, 9, 30}, 0.0933825}, {{1843, 10, 31}, 0.0943574}, {{1843, 11, 30}, 0.0438172}}, {{{1843, 12, 31}, 0.}}}

In[•]:= Length /@ %

Out[•]= {1, 1, 3, 2, 1, 99, 1}
```

By selecting the intervals with values > 0 we collect the draw down intervals.

```
In[∘]:= vxDDIntervals[All] = Select[vxIntervals[All], #⟦1, 2⟧ > 0 &];
```

Once the draw down intervals are available we construct a summary data structure containing the length and depth of each interval.

```
In[∘]:= Length /@ %
Out[∘]= {1, 2, 99, 1, 1, 6, 4, 16, 32, 2, 1, 1, 1, 8, 92, 19, 1, 1, 1, 1, 1, 29, 8, 1, 1, 1, 1, 8, 7, 3, 3, 6, 22, 8, 37, 4, 1, 14, 37, 2, 3,
 4, 24, 2, 6, 15, 1, 1, 1, 2, 54, 7, 1, 3, 2, 11, 1, 9, 1, 25, 4, 1, 6, 26, 2, 1, 17, 9, 34, 5, 1, 29, 1, 2, 29, 2, 1, 3, 10,
 4, 1, 2, 5, 2, 1, 1, 1, 2, 2, 2, 1, 185, 1, 1, 2, 40, 3, 1, 2, 2, 1, 2, 3, 12, 1, 1, 1, 1, 1, 1, 2, 11, 11, 1, 4, 11, 1, 1, 15,
 2, 1, 1, 1, 1, 1, 4, 1, 13, 2, 1, 2, 3, 2, 27, 7, 2, 1, 41, 2, 2, 18, 6, 1, 2, 4, 22, 1, 4, 8, 1, 1, 3, 1, 1, 4, 2, 20, 1, 3,
 4, 8, 2, 3, 3, 1, 3, 1, 1, 1, 1, 6, 4, 1, 2, 1, 1, 3, 1, 4, 1, 1, 4, 2, 4, 73, 2, 3, 52, 4, 3, 1, 1, 1, 1, 1, 2, 2, 1, 9, 1, 3}
```

```
In[∘]:= vxMaxDrawDowns[All] = Transpose[{Length /@ vxDDIntervals[All], Max /@ (Last /@ # & /@ vxDDIntervals[All])}];
```

```
In[∘]:= ListPlot[vxMaxDrawDowns[All], PlotRange → All, PlotStyle → {PointSize[Medium]}, Frame → True, ImageSize → 500,
 FrameLabel → {
   {"Depth", ""},
   {
    "Width",
    Column[
     {
      Style["Market Draw Downs " <> sFromDate <> " through " <> sUntilDate, FontSize → 14],
      Style["S&P 500 Total Return Index", FontSize → 14],
      Style["Source: Global Financial Data", FontSize → 10]
     },
     Center
    ]
   }
  }
]
```

Out[•]=



Modeling Max Draw Downs

Various theoretical and experimental results suggest that the depth of max draw downs for stationary i.i.d random returns will be exponentially distributed. One approach is to assume that the rate parameter λ of the Exponential Distribution is itself a random variable. Here we assume $\lambda \approx$ Gamma Distribution with hyperparameters α and β. We can then integrate out the λ to realize the final distribution:

$$f[x] = \int_0^\infty \left(\frac{\beta^{-\alpha}\, \lambda^{\alpha-1}\, e^{-\lambda/\beta}}{\Gamma[\alpha]} \right) \left(\lambda\, e^{-\lambda\, x} \right) d\lambda = \alpha\, \beta (1 + \beta\, x)^{-(1+\alpha)}$$

In[•]:=
```
Integrate[Evaluate[PDF[GammaDistribution[a, b], l] × PDF[ExponentialDistribution[l], x]],
 {l, 0, ∞}, Assumptions → a > 1 ∧ b > 0]
```

Out[•]=

$$\begin{cases} a\, b\, (1 + b\, x)^{-1-a} & x \geq 0 \\ 0 & \text{True} \end{cases}$$

This distribution is called the Lomax distribution which is in the Pareto family. It can also be represented by the Beta Prime Distribution with parameters noted below.

```
In[]:= FullSimplify[PDF[BetaPrimeDistribution[1, a, 1, 1/b], x], Assumptions → Assumptions → a > 1 ∧ b > 0]
```

$$\text{Out[]=} \begin{cases} a\, b\, (1 + b\, x)^{-1-a} & x > 0 \\ 0 & \text{True} \end{cases}$$

Based on this analysis we can explore a distribution fit on the draw down depths (*i.e.*, max draw downs). Note that we are using the form of the BetaPrimeDistribution used in *Mathematica* and the second parameter is the reciprocal of the β we derived above for the Lomax distribution.

```
In[]:= distDDSize[All] = EstimatedDistribution[vxMaxDrawDowns[All][[All, 2]], BetaPrimeDistribution[1, a, 1, b]]
Out[]= BetaPrimeDistribution[1, 1.84582, 1, 0.0752978]
```

```
In[]:= hypDDSize[All] = DistributionFitTest[vxMaxDrawDowns[All][[All, 2]], distDDSize[All], {"PValue", "ShortTestConclusion"}]
Out[]= {0.860858, Do not reject}
```

In[•]:=
```
Show[
 Histogram[vxMaxDrawDowns[All][[All, 2]], Automatic, PDF],
 Plot[Evaluate[PDF[distDDSize[All], x]], {x, 0, Max[vxMaxDrawDowns[All][[All, 2]]]}, PlotRange → All],
 ImageSize → 500
]
```

Out[•]=


Given the depth has a Pareto-like distribution, it is reasonable to use the Zipf Distribution, which is the discrete equivalent, to model the length of the draw downs.

In[•]:=
```
distDDLeng[All] = EstimatedDistribution[vxMaxDrawDowns[All][[All, 1]], ZipfDistribution[n, r]]
```

Out[•]= ZipfDistribution[185, 0.540065]

In[•]:=
```
hypDDLeng[All] = DistributionFitTest[vxMaxDrawDowns[All][[All, 1]], distDDLeng[All], {"PValue", "ShortTestConclusion"}]
```

Out[•]= {0.528404, Do not reject}

```
In[•]:= Show[
  Histogram[vxMaxDrawDowns[All][[All, 1]], Automatic, PDF],
  DiscretePlot[Evaluate[PDF[distDDLeng[All], x]], {x, 1, Max[vxMaxDrawDowns[All][[All, 1]]]}, PlotRange → All],
  ImageSize → 500
 ]
```

Out[•]=



Drawdown — Excluding the Great Depression and World War II

Selecting a Subset

A natural question at this point is how the events surrounding the Great Depression, which we will extend to include World War II and its following deep recession, affect the model. Is the Great Depression a distortion, an outlier?

In[∘]:=
```
Show[
 DateListPlot[
  Flatten[#, 1] & /@ {Select[vxIntervals[All], First@First@Last[#] < 1928 &],
    Select[vxIntervals[All], First@First@First[#] > 1950 &]},
  Joined → True,
  PlotRange → All,
  PlotStyle → {Black},
  Filling → Bottom,
  FillingStyle → {Opacity[0.5, Red]},
  GridLines → {Table[{i, 1, 1}, {i, 1840, 2000, 10}], Table[i, {i, 0, 2, 0.1}]},
  FrameLabel → {
    {"Log Draw Down", ""},
    {
     "Date",
     Column[
      {
       Style[
        "Market Draw Downs " <> sFromDate <> " through " <> sUntilDate <> " Excluding GD & WW II", FontSize → 14],
       Style["S&P 500 Total Return Index", FontSize → 14],
       Style["Source: Global Financial Data", FontSize → 10]
      },
      Center
     ]
    }
   }
 ],
 ImageSize → 500,
 ImageMargins → 10
]
```

Out[ ]=



We will define a “non GD and WW II” data set as intervals earlier than 1928 or later than 1950.

In[ ]:=
```
vxIntervals["NonGDWWII"] = Join[Select[vxIntervals[All], First@First@Last[#] < 1928 &],
   Select[vxIntervals[All], First@First@First[#] > 1950 &]];
```

By selecting the intervals with values > 0 we collect the draw down intervals.

In[ ]:=
```
vxDDIntervals["NonGDWWII"] = Select[vxIntervals["NonGDWWII"], #[[1, 2]] > 0 &];
```

Once the draw down intervals are available we construct a summary data structure containing the length and depth of each interval.

In[ ]:=
```
vxMaxDrawDowns["NonGDWWII"] =
  Transpose[{Length /@ vxDDIntervals["NonGDWWII"], Max /@ (Last /@ # & /@ vxDDIntervals["NonGDWWII"])}];
```

```
In[•]:= ListPlot[vxMaxDrawDowns["NonGDWWII"], PlotRange → All,
  PlotStyle → {PointSize[Medium]}, Frame → True, ImageSize → 500,
  FrameLabel → {
    {"Depth", ""},
    {
     "Width",
     Column[
      {
       Style[
        "Market Draw Downs " <> sFromDate <> " through " <> sUntilDate <> " Excluding GD and WW II", FontSize → 14],
       Style["S&P 500 Total Return Index", FontSize → 14],
       Style["Source: Global Financial Data", FontSize → 10]
      },
      Center
     ]
    }
   }
]
```

Out[∘]=



Modeling Max Draw Downs

Based on this analysis we can explore a distribution fit on the draw down depths (*i.e.*, max draw downs).

In[∘]:=
```
distDDSize["NonGDWWII"] =
  EstimatedDistribution[vxMaxDrawDowns["NonGDWWII"][[All, 2]], BetaPrimeDistribution[1, a, 1, b]]
```

Out[∘]= BetaPrimeDistribution[1, 2.01121, 1, 0.083007]

In[∘]:=
```
hypDDSize["NonGDWWII"] = DistributionFitTest[
   vxMaxDrawDowns["NonGDWWII"][[All, 2]], distDDSize["NonGDWWII"], {"PValue", "ShortTestConclusion"}]
```

Out[∘]= {0.918282, Do not reject}

```
In[•]:= Show[
  Histogram[vxMaxDrawDowns["NonGDWWII"][[All, 2]], Automatic, PDF],
  Plot[Evaluate[PDF[distDDSize["NonGDWWII"], x]], {x, 0, Max[vxMaxDrawDowns["NonGDWWII"][[All, 2]]]}, PlotRange → All],
  ImageSize → 500
 ]
```

Out[•]=



Given the depth has a Pareto-like distribution, it is reasonable to use the Zipf Distribution, which is the discrete equivalent, to model the length of the draw downs.

```
In[•]:= distDDLeng["NonGDWWII"] = EstimatedDistribution[vxMaxDrawDowns["NonGDWWII"][[All, 1]], ZipfDistribution[n, r]]
Out[•]= ZipfDistribution[99, 0.497322]
```

```
In[•]:= hypDDLeng["NonGDWWII"] = DistributionFitTest[
    vxMaxDrawDowns["NonGDWWII"][[All, 1]], distDDLeng["NonGDWWII"], {"PValue", "ShortTestConclusion"}]
Out[•]= {0.374859, Do not reject}
```

In[•]:=
```
Show[
 Histogram[vxMaxDrawDowns["NonGDWWII"]⟦All, 1⟧, Automatic, PDF], DiscretePlot[
  Evaluate[PDF[distDDLeng["NonGDWWII"], x]], {x, 1, Max[vxMaxDrawDowns["NonGDWWII"]⟦All, 1⟧]}, PlotRange → All],
 ImageSize → 500
]
```

Out[•]=


Comparisons

Comparing the PDFs of the full data set and the non-GD-WWII data set. The projections of the two analyses are different, but not very different. Clearly, even without observing a large event such as the Great Depression, the possibility of such an event is not insignificant.

In[•]:=
```
Plot[{Evaluate[PDF[distDDSize[All], x]], Evaluate[PDF[distDDSize["NonGDWWII"], x]]},
 {x, 0, Max[vxMaxDrawDowns[All][[All, 2]]]}, PlotRange → All]
```

Out[•]=



In[•]:=
```
Plot[{Evaluate[PDF[distDDSize[All], x]], Evaluate[PDF[distDDSize["NonGDWWII"], x]]},
 {x, 0.5, Max[vxMaxDrawDowns[All][[All, 2]]]}, PlotRange → All]
```

Out[•]=